# MATH 149

LABORATORY ASSIGNMENT 4

# IMPLICIT DIFFERENTIATION

The function $f(x)$ is said to be **implicitly defined** by the equation

(*) E( $x$ , $y$ ) = 0

provided that E( $x$ , $f(x)$ ) = 0 for all $x$ in some interval $I$ .

For example, if E( x , y ) = $x^2 + y^2 - 4$, then $f_1(x) = \sqrt{4 - x^2}$ and $f_2(x) = -\sqrt{4 - x^2}$ both satisfy the equation E( x , y ) = 0, and thus both are implicitly defined by this equation.

It is, however, frequently difficult or impossible to solve an equation of the form E(x,y) = 0 for y in terms of x; this is the case in the example and exercises below. (Try it!) Nevertheless, such an equation may implicitly define one or more functions of x.

It is possible to find the derivative of an implicitly defined function at a given point without first solving the equation (*) for y in terms of x by using a technique called ***implicit differentiation*.** We differentiate both sides of the equation (*) with respect to $x$, viewing $y$ as a function of $x$, applying the usual differentiation rules and ultimately solving for $dy / dx$ in terms of x and y . This procedure can be carried out by using appropriate *Maple* commands.

***Example***

Find the slope, **m**, of the tangent line to the graph of the **cardioid** with equation:

$$x^4 + y^4 - 4\,(y^3 + x^2 + x^2\,y) + 2\,x^2\,y^2 = 0$$ at the point **P** = $\left( \frac{2+\sqrt{3}}{2} \right.$ ,

$\frac{2\sqrt{3}+3}{2}$ ).

First, we enter the equation of the cardioid:

```
> restart:
> eq:=x^4+y^4-4*(y^3+x^2+x^2*y)+2*x^2*y^2=0;
```

$$eq := x^4 + y^4 - 4\,y^3 - 4\,x^2 - 4\,x^2\,y + 2\,x^2\,y^2 = 0$$

Next, we sketch a graph of the cardioid in the coordinate plane using the *Maple* command **implicitplot** which is located in the **plots p**ackage. This part of *Maple* can be accessed by first entering the command **with(plots):** . **(**Note that we ended the command with a colon rather than a semicolon---if you terminate it with a semicolon *Maple* responds with the contents of the package. (Try it!)).

```
> with(plots):
> implicitplot(eq,x=-4..4,y=-1..4,grid=[100,100],color=black);
```




Now we tell *Maple* to treat $y$ as a function of $x$ .

```
> y:=f(x);
```

$$y := \mathrm{f}(x)$$

```
> eq;
```

$$x^4 + f(x)^4 - 4\,f(x)^3 - 4\,x^2 - 4\,x^2\,f(x) + 2\,x^2\,f(x)^2 = 0$$

Observe that Maple has replaced each occurrence of $y$ by f (x) .

To differentiate this equation with respect to $x$, we use the **diff** command.

```
> deq:=diff(eq,x);
```

$$deq := 4\,x^3 + 4\,f(x)^3\left(\frac{d}{dx}f(x)\right) - 12\,f(x)^2\left(\frac{d}{dx}f(x)\right) - 8\,x - 8\,x\,f(x) - 4\,x^2\left(\frac{d}{dx}f(x)\right) + 4\,x\,f(x)^2 + 4\,x^2\,f(x)\left(\frac{d}{dx}f(x)\right) = 0$$

Here Maple uses $\frac{d}{dx}f(x)$ to denote f '( x ), the derivative of f( x ).

We now solve for $\frac{d}{dx}f(x)$ : with a deferred command

```
> Diff(f(x),x)=solve(deq,diff(f(x),x));
```

$$\frac{d}{dx}f(x) = -\frac{x\,(x^2 - 2 - 2\,f(x) + f(x)^2)}{f(x)^3 - 3\,f(x)^2 - x^2 + x^2\,f(x)}$$

The single Maple command "implicitdiff" combines the above computations but uses "y" rather than "f(x)".

```
> y:='y';
```

$$y := y$$

(Previously,we had replaced y by f(x); the command "**y:='y'** "undoes this.)

```
> eq;
```

$$x^4 + y^4 - 4\,y^3 - 4\,x^2 - 4\,x^2\,y + 2\,x^2\,y^2 = 0$$

```
> dydx:=implicitdiff(eq,y,x);
```

$$dydx := -\frac{x\,(x^2 - 2 - 2\,y + y^2)}{y^3 - 3\,y^2 - x^2 + x^2\,y}$$

The point **P** has coordinates $x = \frac{2 + \sqrt{3}}{2}$ and $y = \frac{2\sqrt{3} + 3}{2}$, so in order to find **m**, the slope of the tangent at **P,** we must replace y by $\frac{2\sqrt{3} + 3}{2}$ and $x$ by $\frac{2 + \sqrt{3}}{2}$

in the formula for dy/dx.

```
> subs(y=(2*sqrt(3)+3)/2,x=(2+sqrt(3))/2,dydx);
```

$$-\frac{\left(1+\frac{\sqrt{3}}{2}\right)\left(\left(1+\frac{\sqrt{3}}{2}\right)^2-5-2\sqrt{3}+\left(\sqrt{3}+\frac{3}{2}\right)^2\right)}{\left(\sqrt{3}+\frac{3}{2}\right)^3-3\left(\sqrt{3}+\frac{3}{2}\right)^2-\left(1+\frac{\sqrt{3}}{2}\right)^2+\left(1+\frac{\sqrt{3}}{2}\right)^2\left(\sqrt{3}+\frac{3}{2}\right)}$$

Simplifying this complicated fraction, we find the slope **m** of the tangent line to the graph at **P** to be:

```
> m:=simplify(%);
```

$$m := -1$$

In order to get a decimal representation for **m** accurate to ten significant digits, we may use the **evalf** command.

```
> evalf(m);
```

$$-1.$$

The equation of the tangent is $y = \mathbf{m}\left(x - \frac{2+\sqrt{3}}{2}\right) + \frac{2\sqrt{3}+3}{2}$. We can plot the cardioid and its tangent on the same set of axes as follows:

```
> A:=implicitplot(eq,x=-4..4,y=-1..4,grid=[100,100],color=black):
> B:=plot(m*(x-(2+sqrt(3))/2)+(2*sqrt(3)+3)/2,x=-4..4,y=-1..5):
```

**NOTE THAT THE ABOVE COMMANDS END WITH A COLON ":"**
**AND NOT A SEMI-COLON ";".**
**THIS IS DONE TO SUPPRESS THE OUTPUT OF THE COMMAND.**

```
> display([A,B]);
```



5
4
3
y
2
1
0
-1
-4
-3
-2
-1
1
2
3
4
x


>